# 表計算ソフトを用いたコンピュータグラフィックス入門

澤見英男

この資料は，３次元空間における座標変換について学び，表計算ソフト「エクセル」の計算／作図機能を用いた３Ｄ－ＣＧの簡単な演習を通して，膨大なオブジェクトから構成されているＣＧの一端についての理解を深めるべく用意されたものである．

## １　２Ｄ－ＣＧの作図例

座標の回転および平行移動については（２）３Ｄ－ＣＧの基礎知識にて扱うので，ここではエクセルを用いた作図例を取り上げ，計算機を用いた製図法・・・２次元座標上のコンピュータグラフィックスの一端を実感してみることにする．エクセルを立ち上げ，以下のワークシート表示例を参考にしながら，作図作業に慣れるための練習から始めよう．

| | A | B | C | D | E | F | G | H | I |
|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 半径 R | | | | |
| 2 | | | | 3.141593 | 5 | | | | |
| 3 | θ（度） | x=Rcos(θ) | y=Rsin(θ) | | | | 半径5の円を36角形として描く | | |
| 4 | | | | | | | | | |
| 5 | | | | | | | | | |
| 6 | | | | | | | | | |

図１　各種定数とコメントなどの入力

絶対参照し用いるセル$D$2 は式=pi()または円周率の値 3.14159265358979323846 を，セル$E$2 には半径 R の値 5 を入力する．次に，相対参照するセル A4 には角度θ(工学単位)の初期値 0 を，セル A5 には式=A4+10 を入力する．相対参照セル A5 をポインタが太い十文字になるようマウスで指示し，左ボタンを抑えたままで下向きにドラッグすると，以下のような結果が得られる．角度θが 360 になるまで行う．絶対／相対参照は行とか列の前の記号＄の有／無で識別できることから，以降では，絶対／相対参照の説明を略する．

A5　=A4+10

| | A | B | C | D | E | F | G | H | I |
|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 半径 R | | | | 青色の数 |
| 2 | | | | 3.141593 | 5 | | | | |
| 3 | θ（度） | x=Rcos(θ) | y=Rsin(θ) | | | | 半径5の円を36角形として描く | | |
| 4 | 0 | | | | | | x=Rcos(θ) | | |
| 5 | 10 | | | | | | y=Rsin(θ) | | |
| 6 | 20 | | | | | | | | |
| 7 | 30 | | | | | | | | |
| 8 | 40 | | | | | | | | |
| 9 | 50 | | | | | | | | |
| 10 | 60 | | | | | | | | |

図２　作図する範囲を角度で指定

円周を 36 分割する頂点の x 座標に対応するセル B5 は式=$E$2*COS($A4/180*$D$2), y 座標に対応するセル C5 は式=$E$2*SIN($A4/180*$D$2)により，座標値を計算する．座標の値が正しく求まったことを確認してからセル B4 と C5 を選択する．ポインタが太い十文字になるようマウスで指示し，左ボタンを抑えたまま下向きにドラッグすることで，自動計算機能により，以下のような結果を得ることができる．角度 θ が 360 になるまで行う．




| | A | B | C | D | E | F | G | H | I |
|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 半径 R | | | | 青色の数値 |
| 2 | | | | 3.141593 | 5 | | | | |
| 3 | θ（度） | x=Rcos(θ) | y=Rsin(θ) | | | | 半径5の円を36角形として描く | | |
| 4 | 0 | 5 | 0 | | | | x=Rcos(θ) | | |
| 5 | 10 | 4.9240388 | 0.8682409 | | | | y=Rsin(θ) | | |
| 6 | 20 | 4.6984631 | 1.7101007 | | | | | | |
| 7 | 30 | 4.330127 | 2.5 | | | | | | |
| 8 | 40 | 3.8302222 | 3.213938 | | | | | | |
| 9 | 50 | 3.213938 | 3.8302222 | | | | | | |
| 10 | 60 | 2.5 | 4.330127 | | | | | | |

図３　絶対参照と相対参照を使い分けた座標の自動計算

上の例では，セルＢ５の値は５（=5*cos(0)）そしてセルＣ５の値は０（=5*sin(0)）などと正しく計算されている．必要な範囲で正しく計算のできていることを確認したら，以下に示すように，作図に用いる頂点のｘｙ座標を選択する．次に挿入リボンを選択し，その中の散布図プルダウンメニューから散布図（直線）を選ぶことにする．




図４　作図メニューの中から散布図を選択

こうして得られた結果に手を加えることにより，以下のような円を描くことができる．グラフの体裁については，大まかな形はマウスで整え，その他についてはキィワード「凡例，座標軸など」を用いヘルプを使って調べ，好みのものになるよう仕上げると良い．



| | A | B | C | D | E | F | G | H | I |
|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 半径 R | | | | 青色の数 |
| 2 | | | | 3.141593 | 5 | | | | |
| 3 | θ(度) | x=Rcos(θ) | y=Rsin(θ) | | | | 半径5の円を36角形として描く | | |
| 4 | 0 | 5 | 0 | | | | | | |
| 5 | 10 | 4.9240388 | 0.8682409 | | | | | | |
| 6 | 20 | 4.6984631 | 1.7101007 | | | | | | |
| 7 | 30 | 4.330127 | 2.5 | | | | | | |
| 8 | 40 | 3.8302222 | 3.213938 | | | | | | |
| 9 | 50 | 3.213938 | 3.8302222 | | | | | | |
| 10 | 60 | 2.5 | 4.330127 | | | | | | |
| 11 | 70 | 1.7101007 | 4.6984631 | | | | | | |
| 12 | 80 | 0.8682409 | 4.9240388 | | | | | | |
| 13 | 90 | 8.078E-15 | 5 | | | | | | |
| 14 | 100 | -0.868241 | 4.9240388 | | | | | | |
| 15 | 110 | -1.710101 | 4.6984631 | | | | | | |
| 16 | 120 | -2.5 | 4.330127 | | | | | | |
| 17 | 130 | -3.213938 | 3.8302222 | | | | | | |
| 18 | 140 | -3.830222 | 3.213938 | | | | | | |
| 19 | 150 | -4.330127 | 2.5 | | | | | | |
| 20 | 160 | -4.698463 | 1.7101007 | | | | | | |
| 21 | 170 | -4.924039 | 0.8682409 | | | | | | |
| 22 | 180 | -5 | 1.616E-14 | | | | | | |
| 23 | 190 | -4.924039 | -0.868241 | | | | | | |
| 24 | 200 | -4.698463 | -1.710101 | | | | | | |
| 25 | 210 | -4.330127 | -2.5 | | | | | | |



図５　縦軸と横軸のスケールを調整して描画

必要な個数だけ頂点のｘｙ座標を求め，幾何的な図形を描くことができる．例として，螺旋状の図形とスピログラフ様の描画例を示す．座標を計算する式だけが異なっている．



| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | | R | | | 青色の数値を変更する | |
| 2 | | | | 3.141593 | | 10 | | | | |
| 3 | θ(度) | x | y | | | | | | | |
| 4 | 0 | 0 | 0 | | R=10とし，5度刻みで螺旋を描く | | | | | |
| 5 | 5 | 0.2767207 | 0.0242099 | | x=R*(θ/180)*cos(θ*π/180) | | | | | |
| 6 | 10 | 0.5471154 | 0.0964712 | | y=R*(θ/180)*sin(θ*π/180) | | | | | |
| 7 | 15 | 0.8049382 | 0.2156825 | | | | | | | |
| 8 | 20 | 1.0441029 | 0.3800224 | | | | | | | |
| 9 | 25 | 1.2587608 | 0.5869698 | | | | | | | |
| 10 | 30 | 1.4433757 | 0.8333333 | | | | | | | |
| 11 | 35 | 1.5927956 | 1.1152875 | | | | | | | |
| 12 | 40 | 1.702321 | 1.4284169 | | | | | | | |
| 13 | 45 | 1.767767 | 1.767767 | | | | | | | |
| 14 | 50 | 1.7855211 | 2.1279012 | | | | | | | |
| 15 | 55 | 1.7525947 | 2.5029646 | | | | | | | |
| 16 | 60 | 1.6666667 | 2.8867513 | | | | | | | |
| 17 | 65 | 1.5261215 | 3.2727781 | | | | | | | |
| 18 | 70 | 1.3300783 | 3.6543602 | | | | | | | |
| 19 | 75 | 1.0784127 | 4.0246909 | | | | | | | |
| 20 | 80 | 0.7717697 | 4.3769233 | | | | | | | |
| 21 | 85 | 0.4115688 | 4.7042527 | | | | | | | |
| 22 | 90 | 8.078E-15 | 5 | | | | | | | |
| 23 | 95 | -0.459989 | 5.2576942 | | | | | | | |
| 24 | 100 | -0.964712 | 5.4711542 | | | | | | | |



図６　螺旋の描画



B4 =$F$2*COS($A4/180*$D$2)+$G$2*COS($E$2*$A4/180*$D$2)

| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 倍率 α | 半径 R1 | 半径 R2 | | 青色の数値を変更する | |
| 2 | | | | 3.141593 | 5 | 10 | 6 | | | |
| 3 | θ(度) | x | y | | | | | | | |
| 4 | 0 | 16 | 0 | | 半径R1の円周の軌跡と半径R2でα倍早く回る円周の軌跡との合成 | | | | | |
| 5 | 5 | 15.399794 | 3.407267 | | x=R1*cos(θ*π/180)+R2*cos(α*θ*π/180) | | | | | |
| 6 | 10 | 13.704803 | 6.3327484 | | y=R1*sin(θ*π/180)+R2*sin(α*θ*π/180) | | | | | |
| 7 | 15 | 11.212173 | 8.3837454 | | | | | | | |
| 8 | 20 | 8.3550371 | 9.329048 | | | | | | | |
| 9 | 25 | 5.6216193 | 9.1410949 | | | | | | | |
| 10 | 30 | 3.4641016 | 8 | | | | | | | |
| 11 | 35 | 2.2143523 | 6.2586988 | | | | | | | |
| 12 | 40 | 2.0222887 | 4.3757552 | | | | | | | |
| 13 | 45 | 2.8284271 | 2.8284271 | | | | | | | |
| 14 | 50 | 4.3757552 | 2.0222887 | | | | | | | |
| 15 | 55 | 6.2586988 | 2.2143523 | | | | | | | |
| 16 | 60 | 8 | 3.4641016 | | | | | | | |
| 17 | 65 | 9.1410949 | 5.6216193 | | | | | | | |
| 18 | 70 | 9.329048 | 8.3550371 | | | | | | | |
| 19 | 75 | 8.3837454 | 11.212173 | | | | | | | |
| 20 | 80 | 6.3327484 | 13.704803 | | | | | | | |
| 21 | 85 | 3.407267 | 15.399794 | | | | | | | |
| 22 | 90 | 6.596E-14 | 16 | | | | | | | |
| 23 | 95 | -3.407267 | 15.399794 | | | | | | | |
| 24 | 100 | -6.332748 | 13.704803 | | | | | | | |
| 25 | 105 | -8.383745 | 11.212173 | | | | | | | |
| 26 | 110 | -9.329048 | 8.3550371 | | | | | | | |
| 27 | 115 | -9.141095 | 5.6216193 | | | | | | | |
| 28 | 120 | -8 | 3.4641016 | | | | | | | |
| 29 | 125 | -6.258699 | 2.2143523 | | | | | | | |
| 30 | 130 | -4.375755 | 2.0222887 | | | | | | | |
| 31 | 135 | -2.828427 | 2.8284271 | | | | | | | |



図７　スピログラフ

曲線を折線により近似して描画する例を幾つか見てきたが，同じようにして直線によるグラフを描くこともできる．例として，半径Ｒの円周をＮ等分して得られるＮ個の頂点から間隔がｋのものを選び直線で結ぶと，どのような図形が得られるのかを確かめてみよう．表計算ソフトの作図機能を用いた製図作業に親しみを感じてきたのではないだろうか．



B4 =$E$2*COS($A4*$D$2/180)

| | A | B | C | D | E | F | G | H | I | J | K |
|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | 円周率 π | 半径 R | 間隔 k | 分割数 N | | 青色の数値を変更すると図形全体か | | |
| 2 | | | | 3.141593 | 5 | 6 | 17 | | | | |
| 3 | θ(度) | x | y | | | | | | | | |
| 4 | 0 | 5 | 0 | | 半径Rの円周をN等分して得られる点をk個おきに直線で結ぶことにより星形を描く | | | | | | |
| 5 | 127.0588 | -3.01317 | 3.990086 | | x=Rcos(θ*π/180)=Rcos(2πk/N) | | | | | | |
| 6 | 254.1176 | -1.36831 | -4.80913 | | y=Rsin(θ*π/180)=Rsin(2πk/N) | | | | | | |
| 7 | 381.1765 | 4.662361 | 1.806208 | | | | | | | | |
| 8 | 508.2353 | -4.25109 | 2.63216 | | | | | | | | |
| 9 | 635.2941 | 0.461342 | -4.97867 | | | | | | | | |
| 10 | 762.3529 | 3.695045 | 3.368478 | | | | | | | | |
| 11 | 889.4118 | -4.91487 | 0.918748 | | | | | | | | |
| 12 | 1016.471 | 2.228692 | -4.47582 | | | | | | | | |
| 13 | 1143.529 | 2.228692 | 4.475816 | | | | | | | | |
| 14 | 1270.588 | -4.91487 | -0.91875 | | | | | | | | |
| 15 | 1397.647 | 3.695045 | -3.36848 | | | | | | | | |
| 16 | 1524.706 | 0.461342 | 4.978671 | | | | | | | | |
| 17 | 1651.765 | -4.25109 | -2.63216 | | | | | | | | |



図８　多角形または星形の図形

## ２　基本的な座標変換と表計算ソフトについて

ここでは３次元ＣＧの基本，座標の回転操作と平行移動について理解しておこう．先ず，３次元座標における点(X,Y,Z)をｘ軸の周りで角度$\alpha$だけ回転操作して得られる点の座標(x, y, z)は，行列Ｒx とベクトルにより表すと，次のように計算することができる．

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & -\sin\alpha \\ 0 & \sin\alpha & \cos\alpha \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \end{bmatrix} = Rx \begin{bmatrix} X \\ Y \\ Z \end{bmatrix}$$

同様にして，３次元座標における点(X,Y,Z)をｙ軸の周りで$\beta$回転操作して得られる点の座標(x, y, z)を，行列Ｒy を用いて次のように表すことができる．

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} \cos\beta & 0 & \sin\beta \\ 0 & 1 & 0 \\ -\sin\beta & 0 & \cos\beta \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \end{bmatrix} = Ry \begin{bmatrix} X \\ Y \\ Z \end{bmatrix}$$

さらに，３次元座標における点(X,Y,Z)をｚ軸の周りで$\gamma$回転操作して得られる点の座標(x, y, z)を，行列Ｒz を用いて次のように表すことができる．

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} \cos\gamma & -\sin\gamma & 0 \\ \sin\gamma & \cos\gamma & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \end{bmatrix} = Rz \begin{bmatrix} X \\ Y \\ Z \end{bmatrix}$$

これらの結果を用いると，点(X,Y,Z)を最初はｘ軸の周りで$\alpha$回転し，次にｙ軸の周りで$\beta$，最後にｚ軸の周りで$\gamma$回転操作して得られる点の座標(x, y, z)は次のように表すことができる．勿論，３個の行列をひとまとめにしてひとつの行列により表すこともできる．

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} \cos\gamma & -\sin\gamma & 0 \\ \sin\gamma & \cos\gamma & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} \cos\beta & 0 & \sin\beta \\ 0 & 1 & 0 \\ -\sin\beta & 0 & \cos\beta \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & -\sin\alpha \\ 0 & \sin\alpha & \cos\alpha \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \end{bmatrix} = RzRyRx \begin{bmatrix} X \\ Y \\ Z \end{bmatrix}$$

同様にして，３次元座標における点(X, Y, Z)を最初はｚ軸の周りで$\gamma$回転し，次にｙ軸の周りで$\beta$，最後にｘ軸の周りで$\alpha$回転操作して得られる座標(x, y, z)を求めることができる．

$$\begin{bmatrix} x \\ y \\ z \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 \\ 0 & \cos\alpha & -\sin\alpha \\ 0 & \sin\alpha & \cos\alpha \end{bmatrix} \begin{bmatrix} \cos\beta & 0 & \sin\beta \\ 0 & 1 & 0 \\ -\sin\beta & 0 & \cos\beta \end{bmatrix} \begin{bmatrix} \cos\gamma & -\sin\gamma & 0 \\ \sin\gamma & \cos\gamma & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \end{bmatrix} = RxRyRz \begin{bmatrix} X \\ Y \\ Z \end{bmatrix}$$

ところで，点(X, Y, Z)をｘｙｚ軸方向に(a, b, c)平行移動操作した点の座標(x, y, z)を次のように４次元の行列を用いて表すことができる（同次座標系における平行移動）．

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & a \\ 0 & 1 & 0 & b \\ 0 & 0 & 1 & c \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

この同次座標系を用いると，３次元座標における点(X,Y,Z)をｘ軸の周りで$\alpha$回転操作して得られる点の座標(x, y, z)は次のようになる（同次座標系におけるｘ軸周りの回転）．

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & \cos\alpha & -\sin\alpha & 0 \\ 0 & \sin\alpha & \cos\alpha & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

同様に，点(X, Y, Z)をｙ軸またはｚ軸の周りで角度$\beta$または$\gamma$回転操作して得られる点の座標(x, y, z)は次のようになる（同次座標系におけるｙまたはｚ軸周りの回転）.

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} \cos\beta & 0 & \sin\beta & 0 \\ 0 & 1 & 0 & 0 \\ -\sin\beta & 0 & \cos\beta & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} \cos\gamma & -\sin\gamma & 0 & 0 \\ \sin\gamma & \cos\gamma & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

これより，３次元座標における点(X, Y, Z)をｘ軸の周りで角度$\alpha$回転操作した後にｘｙｚ軸方向に(a, b, c)平行移動操作して得られる点の座標(x, y, z)は次のようになる.

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & a \\ 0 & \cos\alpha & -\sin\alpha & b \\ 0 & \sin\alpha & \cos\alpha & c \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

さらに，点(X,Y,Z)をｙ軸またはｚ軸の周りで$\beta$または$\gamma$回転操作した後に，任意の方向(d, e, f)または別の方向(p, q, r)へと平行移動操作した点の座標(x, y, z)は次のようになる.

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} \cos\beta & 0 & \sin\beta & d \\ 0 & 1 & 0 & e \\ -\sin\beta & 0 & \cos\beta & f \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} \cos\gamma & -\sin\gamma & 0 & p \\ \sin\gamma & \cos\gamma & 0 & q \\ 0 & 0 & 1 & r \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

このような行列を用い，軸周りの回転操作とそれに続く平行移動とをひとまとめに表すことができる．ところで，各座標軸周りの回転操作をひとまとめに済ませてからｘｙｚ軸方向(a, b, c)に平行移動するようにすると，回転軸の順序により行列の成分（A11,・・・, A33）の値は異なるが，一般に，以下のような形式で表すことができる．この場合，回転を表す成分は計算手順により変化するものの，平行移動を表す成分はそのままになる.

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} A11 & A12 & A13 & a \\ A21 & A22 & A23 & b \\ A31 & A32 & A33 & c \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

一方，点(X,Y,Z)をｘｙｚ軸方向(a, b, c)に平行移動操作してからｘ軸の周りで角度$\alpha$だ

け回転操作して得られる座標(x, y, z)は次のようになり，演算の順序により結果の異なることが分かる（非可換）．先の結果とは異なり，回転操作によって平行移動成分の値は変化する．

$$\begin{bmatrix} x \\ y \\ z \\ 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & a \\ 0 & \cos\alpha & -\sin\alpha & (b\cos\alpha - c\sin\alpha) \\ 0 & \sin\alpha & \cos\alpha & (b\sin\alpha + c\cos\alpha) \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} X \\ Y \\ Z \\ 1 \end{bmatrix}$$

このような行列およびベクトルを用いた計算を表計算ソフト「エクセル」で行うのには，以降で示すような，絶対参照／相対参照／複合参照による計算式を用いると便利である．


C6 =COS($A$3*PI()/180)

| | A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | 行列(Rx,Ry,Rz)の定義 | | | | | | | | | | | | | |
| 2 | α | | | | | | β | | | | | | γ | | | |
| 3 | 0 | | | | | | 0 | | | | | | 0 | | | |
| 4 | | | | | | | | | | | | | | | | |
| 5 | | 1 | 0 | 0 | | | | 1 | 0 | 0 | | | | 1 | 0 | 0 |
| 6 | Rx:= | 0 | 1 | 0 | | | Ry:= | 0 | 1 | 0 | | | Rz:= | 0 | 1 | 0 |
| 7 | | 0 | 0 | 1 | | | | 0 | 0 | 1 | | | | 0 | 0 | 1 |

図９　行列(Rx, Ry, Rz)を定義する

例として，３次元の回転操作を表す行列(Rx, Ry, Rz)を取り上げる．ワークシート上で，行列 Rx の(1,1)成分を値１に，(1,2)成分と(1,3)成分そして(2,1)成分は値０に設定する．さらに(2,2)成分と(2,3)成分を式「=COS($A$3*PI()/180)，=-SIN($A$3*PI()/180)」，(3,1)成分は値０に，(3,2)成分と(3,3)成分を式「=SIN($A$3*PI()/180)，=COS($A$3*PI()/180)」により設定する（図９）．式中の絶対参照セル「＄Ａ＄３」は，ｘ軸周りの回転角 α を工学単位により表している．工学単位による角度は，理学単位に基づき三角関数(COS, SIN)を計算していることから，係数(π/180）倍する．行列 Ry と Rz についても同様にして設定する．


B12 =$B5*H$5+$C5*H$6+$D5*H$7

| | A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | 行列(Rx,Ry,Rz)の定義 | | | | | | | | | | | | | |
| 2 | α | | | | | | β | | | | | | γ | | | |
| 3 | 45 | | | | | | 45 | | | | | | 45 | | | |
| 4 | | | | | | | | | | | | | | | | |
| 5 | | 1 | 0 | 0 | | | | 0.7071 | 0 | 0.7071 | | | | 0.7071 | -0.707 | 0 |
| 6 | Rx:= | 0 | 0.7071 | -0.707 | | | Ry:= | 0 | 1 | 0 | | | Rz:= | 0.7071 | 0.7071 | 0 |
| 7 | | 0 | 0.7071 | 0.7071 | | | | -0.707 | 0 | 0.7071 | | | | 0 | 0 | 1 |
| 8 | | | | | | | | | | | | | | | | |
| 9 | | | | | | | | | | | | | | | | |
| 10 | | | 行列と行列の積 | | | | | | | | | | | | | |
| 11 | | | | | | | | | | | | | | | | |
| 12 | | 0.7071 | 0 | 0.7071 | | | | 0.7071 | 0.5 | 0.5 | | | | 0.5 | -0.146 | 0.8536 |
| 13 | RxRy:= | 0.5 | 0.7071 | -0.5 | | | RyRx:= | 0 | 0.7071 | -0.707 | | | RzRyRx:= | 0.5 | 0.8536 | -0.146 |
| 14 | | -0.5 | 0.7071 | 0.5 | | | | -0.707 | 0.5 | 0.5 | | | | -0.707 | 0.5 | 0.5 |

図１０　行列の積(RxRy, RyRx, RzRyRx)を計算する

複合参照による式「=$B5*H$5+$C5*H$6+$D5*H$7」により，行列の積(RxRy)の(1,1)成分をセル「Ｂ１２」上に求める．次に，このセル「Ｂ１２」に求めた(1,1)成分を，残る８成分に対応したセルへとコピーし行列の積(RxRy)を得る．同様に，行列の積(RyRx)の(1,1)成分を式「=$H5*B$5+$I5*B$6+$J5*B$7」によりセル「Ｈ１２」に，行列の積(RzRyRx)の(1,1)成分を式「=$N5*H$12+$O5*H$13+$P5*H$14」によりセル「Ｎ１２」に求め，これらを対応する残りのセルにコピーして，行列の積を得る．ここで示した例では，座標軸周りの回転角度α，β，γいずれも４５度になっている（図１０）．このような表計算の機能を用いると，行列とベクトルの積や内積も簡単に計算できることを注記しておく．

## ３　平行投影を用いた簡単な３Ｄ－ＣＧについて

３次元空間における対象物（３Ｄオブジェクト）を２次元平面上に線画として描くには，３Ｄ－２Ｄ変換が必要となる．ここでは簡単のため，３Ｄオブジェクトへの回転操作のみを考え，２Ｄ変換には平行光線による投影（平行投影）を用いることにする．ところで，投影面の定義は２軸で済むが，３Ｄ空間での扱いは３軸を必要とすることを注記しておく．

平行投影では，３Ｄ空間中のオブジェクト（対象物）を平行投影する平面である投影面を，原点とｘ軸およびｙ軸に対応する２本の３Ｄ単位ベクトルにより定義する．次に，３Ｄオブジェクト上の点を表す３Ｄ座標を３Ｄベクトルとして扱い，この３Ｄベクトルと２本の３Ｄ単位ベクトルそれぞれとの内積を求め，この値を投影面上の２Ｄ座標とする．

このように３Ｄ－２Ｄ変換には，（１）回転操作を３Ｄオブジェクト上の点である３Ｄ座標（３Ｄベクトル）に適用してから，得られた３Ｄベクトルと２本の３Ｄ基本ベクトル（例えば **e1** と **e2**）との内積を計算するか，（２）回転操作を２本の３Ｄ単位ベクトルに適用してから，３Ｄオブジェクト上の点でもある３Ｄベクトルとの内積を計算することにより２Ｄ座標を求める方法がある（右手座標系もしくは左手座標系に対応させることができる）．

ここでは簡単のため，視点の移動と向きの変更に対応する，（２）２本の３Ｄ単位ベクトルに座標変換を適用する方法を用いることにする．なお，３Ｄオブジェクト上の点をベクトルで表す際の基準点を平行移動することで，平行移動操作を行うことができる．これらのことから，同次座標系ではなく通常の座標系を用いた作図手順を示すことにする．

例としてｘ軸およびｙ軸に相当する，３Ｄ基本ベクトル **e1** と **e2** を選ぶことにする．

$$\mathbf{e1}=\begin{bmatrix}1\\0\\0\end{bmatrix}\quad,\quad\mathbf{e2}=\begin{bmatrix}0\\1\\0\end{bmatrix}$$

これら２本の基本ベクトルに回転操作を加えることにより，投影面を定義する２本の３Ｄ単位ベクトル **u1** と **u2** を得ることができる．

$$\mathrm{u1}=\begin{bmatrix} u11 \\ u12 \\ u13 \end{bmatrix}\quad,\quad \mathrm{u2}=\begin{bmatrix} u21 \\ u22 \\ u23 \end{bmatrix}$$

３Ｄオブジェクト上の各点を表すベクトルとこれら単位ベクトルとの内積を順次求めることにより，投影面上の２Ｄ座標が決まることから，３Ｄオブジェクト上の各点の連結関係に従って，線画（ワイヤーフレーム）として投影面に３Ｄ対象物を描くことができる．

以下に表計算ソフト「エクセル」を用いた上記の３Ｄ－２Ｄ変換および描画の例を示す．ここでは，３次元空間における対象物として螺旋状に巻かれた針金のようなものを取り上げることにする．例として，３Ｄ対象物上の座標を以下のようにして与えている．

$$\begin{bmatrix} X \\ Y \\ Z \end{bmatrix}=\begin{bmatrix} |1+\sin(\theta/k)|R\cos(\theta) \\ |1+\sin(\theta/k)|R\sin(\theta) \\ s\theta \end{bmatrix}$$

螺旋形１個であれば一筆書きにより描画できることから，エクセルの散布図による描画機能を用い，３Ｄ－ＣＧの雰囲気を十分味わうことができる．以下に作業手順の例を示す．

B5　=ABS(1+SIN($A5*$D$2/180/$J$2))*$E$2*COS($A5*$D$2/180)

| | A | B | C | D | E | F | G | H | I | J | K | L |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | π | R | s | α(z軸回転) | β(y軸回転) | γ(x軸回転) | 変化率k | | |
| 2 | | | | 3.141593 | 10 | 0.03 | 0 | 0 | 0 | 26 | | |
| 3 | 角度 | 3次元空間での座標 | | | 2次元平面での座標 | | | | | | | 1 |
| 4 | θ(度) | x=\|1+sin(θ | y=\|1+sin(θ | z=sθ | X | Y | | | | | | 0 |
| 5 | 0 | 10 | 0 | 0 | 10 | 0 | | | | | | 0 |
| 6 | 10 | 9.914185 | 1.748138 | 0.3 | 9.914185 | 1.748138 | | | | | | |
| 7 | 20 | 9.523082 | 3.466118 | 0.6 | 9.523082 | 3.466118 | | | | 1 | | |
| 8 | 30 | 8.834646 | 5.100685 | 0.9 | 8.834646 | 5.100685 | | | 横軸: u1 = | 0 | <=Rz(Ry(Rxe1)) | |
| 9 | 40 | 7.866112 | 6.600452 | 1.2 | 7.866112 | 6.600452 | | | | 0 | | |
| 10 | 50 | 6.643581 | 7.917511 | 1.5 | 6.643581 | 7.917511 | | | | | | |
| 11 | 60 | 5.20133 | 9.008967 | 1.8 | 5.20133 | 9.008967 | | | | | | |
| 12 | 70 | 3.580856 | 9.838322 | 2.1 | 3.580856 | 9.838322 | | | | 0 | | |
| 13 | 80 | 1.82969 | 10.37669 | 2.4 | 1.82969 | 10.37669 | | | 縦軸: u2= | 1 | <=Rz(Ry(Rxe2)) | |
| 14 | 90 | 1.71E-14 | 10.60378 | 2.7 | 1.71E-14 | 10.60378 | | | | 0 | | |
| 15 | 100 | -1.85296 | 10.50866 | 3 | -1.85296 | 10.50866 | | | | | | |
| 16 | 110 | -3.67252 | 10.09017 | 3.3 | -3.67252 | 10.09017 | | x=\|1+sin(θ/k)\|Rcosθ | | | | |
| 17 | 120 | -5.40233 | 9.357115 | 3.6 | -5.40233 | 9.357115 | | y=\|1+sin(θ/k)\|Rsinθ | | | | |
| 18 | 130 | -6.9881 | 8.328096 | 3.9 | -6.9881 | 8.328096 | | z=sθ | | | | |

図１１　３Ｄオブジェクトの座標計算

セル「Ａ５」には開始角度０を与えておく．そして，セル「Ａ６」には計算式「＝Ａ５＋１０」を入力し，自動計算により値が１０００程度になるよう下向きにドラッグする．セル「Ｂ５」ではｘ座標値 $|1+\sin(\theta/k)|R\cos\theta$ を計算する．角度 $\theta$ はセル「＄Ａ５」に与えられており，この例では０度（工学単位）となっているので，セル「＄Ｄ＄２」の円周率（１８０度に相当）を用いて理学単位に直している．そして，何回転で元に戻るかを表す変数「ｋ」を変化率としセル「＄Ｊ＄２］に指定する．この例では２６となる．半径Ｒはセル「＄Ｅ＄２」に指定する．同様にしてセル「Ｃ５」ではｙ座標値 $|1+\sin(\theta/k)|R\sin\theta$ を計算している．セル「Ｄ５」では工学単位による角度とスケール因子ｓを用い式「=$A5*$F$2」によりｚ座標値の値 $s\theta$ を計算するようにしている．

計算できていることを確認したら，３Ｄ座標「セルＢ５からＤ５まで」を選び，下にドラッグしてコピーすることで作図に必要な範囲に関する自動計算を行うようにする．これ

までの操作により，３Ｄオブジェクトの座標を全て得ることができる．次は２本の３Ｄ基本ベクトル e1 と e2 をセル「Ｖ７～Ｖ９」とセル「Ｖ１２～Ｖ１３」に設定しておく．




図１２　２本の３Ｄ基本ベクトル e1 と e2

基本ベクトル e1 と e2 を行列Ｒｘによりｘ軸の周りでγ度回転させる．行列Ｒｘの成分はセル「Ｔ３～Ｖ６」に設定する．行列Ｒｘの２行２列目の成分は式「=COS($D$2*$I$2/180)」により計算する．ただし，セル「＄Ｉ＄２」は回転角度γである．同様にして行列Ｒｘの２行３列目の成分（=-SIN($D$2*$I$2/180)），３行２列目の成分（=SIN($D$2*$I$2/180)）そして３行３列目の成分（=COS($D$2*$I$2/180)）も計算するようにしておく．




図１３　行列と基本ベクトルとの積を計算する

次に行列(Rx)と２本の基本ベクトル(e1 , e2)との積を求める．ベクトル(Rx)e1 のｘ成分「=$T$3*V$7+$U$3*V$8+$V$3*V$9)，ｙ成分「=$T$4*V$7+$U$4*V$8+$V$4*V$9」，ｚ成分「=$T$5*V$7+$U$5*V$8+$V$5*V$9」を計算する．同様にして(Rx) e2 のｘｙｚ成分「=$T$3*V$12+$U$3*V$13+$V$3*V$14　，=$T$4*V$12+$U$4*V$13+$V$4*V$14　，=$T$5*V$12+$U$5*V$13+$V$5*V$14」も計

算しておく．行列とベクトルの積を求めるため，ベクトル成分３個についての計算式をセルの絶対参照を用い書き表している．

先に示したように，成分ひとつだけを式により計算し，残りは自動計算により求めることもできる．例えば，ベクトル(Rx)e1 の x 成分を式「=$T3*V$7+$U3*V$8+$V3*V$9)」により与え，残る y 成分と z 成分はこの計算式をコピーして済ませることもできる．同様に (Rx)e2 の x 成分を複合参照による式「=$T3*V$12+$U3*V$13+$V3*V$14」により計算するようにし，残り２個の y z 成分はコピーによる自動計算で済ますことができる．

ところで，ベクトル成分の計算式を全て相対参照により書き表すこともできる．この場合，例えばベクトル(Rx)e1 の各成分「=T3*V7+U3*V8+V3*V9)，=T4*V7+U4*V8+V4*V9，=T5*V7+U5*V8+V5*V9」の計算式は随分とすっきりした感じになる．しかし，行列やベクトルを用いた計算をこのような相対参照だけによる計算式で行おうとすると，この式をコピーするだけではなく，正しい式へと書き換えるための手間が必要になる．




図１４　投影面上の単位ベクトル **u1** と **u2** を求める

基本ベクトル **e1** および **e2** と回転操作に対応する行列(Rx, Ry, Rz)との積を順次求めることにより，投影面上の単位ベクトル **u1** および **u2** を計算した．上に示した例では，回転角度が全てゼロであることから **u1 = e1** および **u2= e2** となっている．正しく計算できていることを確かめるには，例えば，各回転角度として 45(度)とか 90(度)さらには 180(度)などの値を入力し，行列(Rx, Ry, Rz)の各成分や基本ベクトル **e1** および **e2** との積がどうなるのかを見ると良いであろう．これで３Ｄ－２Ｄ変換の準備が整ったことになる．

３Ｄ対象物上の点を表す３Ｄベクトルと，投影面を表す基本ベクトル(u1, u2)との内積は，投影面上の２Ｄ座標に対応する．したがって，このような３Ｄ－２Ｄ変換を順次行い，得られた２Ｄ座標を使い作図機能を用いて，平行投影による図を描画することができる．

SUM　=$B5*$J$7+$C5*$J$8+$D5*$J$9

| | A | B | C | D | E | F | G | H | I | J |
|---|---|---|---|---|---|---|---|---|---|---|
| 1 | ラジアン=θ×(π/180) | | | π | R | s | α(z軸回転) | β(y軸回転) | γ(x軸回転) | 変化率k |
| 2 | | | | 3.141593 | 10 | 0.03 | 0 | 0 | 0 | 26 |
| 3 | 角度 | 3次元空間での座標 | | | 2次元平面での座標 | | | | | |
| 4 | θ(度) | x=\|1+sin(θ | y=\|1+sin(θ | z=sθ | X | Y | | | | |
| 5 | 0 | 10 | 0 | 0 | =$B5*$J$7+$C5*$J$8+$D5*$J$9 | | | | | |
| 6 | 10 | 9.914185 | 1.748138 | 0.3 | 9.914185 | 1.748138 | | | | |
| 7 | 20 | 9.523082 | 3.466118 | 0.6 | 9.523082 | 3.466118 | | | | 1 |
| 8 | 30 | 8.834646 | 5.100685 | 0.9 | 8.834646 | 5.100685 | | | 横軸: u1 = | 0 |
| 9 | 40 | 7.866112 | 6.600452 | 1.2 | 7.866112 | 6.600452 | | | | 0 |
| 10 | 50 | 6.643581 | 7.917511 | 1.5 | 6.643581 | 7.917511 | | | | |
| 11 | 60 | 5.20133 | 9.008967 | 1.8 | 5.20133 | 9.008967 | | | | |
| 12 | 70 | 3.580856 | 9.838322 | 2.1 | 3.580856 | 9.838322 | | | | 0 |
| 13 | 80 | 1.82969 | 10.37669 | 2.4 | 1.82969 | 10.37669 | | | 縦軸: u2= | 1 |
| 14 | 90 | 1.71E-14 | 10.60378 | 2.7 | 1.71E-14 | 10.60378 | | | | 0 |
| 15 | 100 | -1.85296 | 10.50866 | 3 | -1.85296 | 10.50866 | | | | |
| 16 | 110 | -3.67252 | 10.09017 | 3.3 | -3.67252 | 10.09017 | | x=\|1+sin(θ/k)\|Rcosθ | | |
| 17 | 120 | -5.40233 | 9.357115 | 3.6 | -5.40233 | 9.357115 | | y=\|1+sin(θ/k)\|Rsinθ | | |
| 18 | 130 | -6.9881 | 8.328096 | 3.9 | -6.9881 | 8.328096 | | z=sθ | | |

図１５　基本ベクトルと３Ｄ対象物上の点を表すベクトルとの内積

上に示した例では，３Ｄ空間中の点を表すセル「Ｂ５～Ｄ５」と基本ベクトル **u2** を表すセル「Ｊ７～Ｊ９」を用い，２Ｄ平面上でのｘ座標に相当するセル「Ｅ５」をベクトルの内積を表す計算式「=$B5*$J$7+$C5*$J$8+$D5*$J$9」により求めている．ｙ座標に相当するセル「Ｆ５」は式「=$B5*$J$12+$C5*$J$13+$D5*$J$14」により計算している．計算式を，投影面上の単位ベクトル **u1** および **u2** を絶対参照で，３Ｄ対象物上の点を表すベクトルを絶対参照と相対参照を交えた複合参照で書き表している．また３Ｄベクトルは全て相対参照することもできる．しかし，複合参照により，セルの再利用は容易になる．

角度θは１０度刻みで与えることにする．先に述べた式により，３Ｄ対象物上の点をベクトルとして列「Ｂ～Ｄ」に自動計算により求める．平行投影に基づく３Ｄ－２Ｄ変換により，列「Ｅ～Ｆ」に３Ｄ投影面上の２Ｄ座標値を求めておく．こうして得られた２Ｄ座標値を選択し，散布図として３Ｄ対象物の平行投影図を描くことができる（図１６）．



図１６　上から見た半径の変化する回転螺旋形($\alpha=\beta=\gamma=0$)

この３Ｄ対象物はｚ軸周りで半径を変えながら回転させている螺旋形なので，試しにスケール因子ｓ=0.03 とｓ=0.02 について，ｘ軸周りに＋１０度ｙ軸周りに－１０度回転させてみることにする．これは，投影面をｘ軸周りに－１０度ｙ軸周りに＋１０度回転させたことに相当する．これは，ここで示した３Ｄ－２Ｄ変換に左手座標系を用いていることを意味している．この例では，主な計算は３Ｄベクトルの内積になっており，従って計算機への負荷の軽い計算量の少ない手法になっている（図１５）．

右手座標系を用いて作図するためには，３Ｄ対象物上の全ての点を表す３Ｄベクトルに対し行列(Rx, Ry, Rz)を用いた回転処理を行う必要がある．そのためには，計算量を減らすべく回転操作を表す３個の行列の積を予め求めておき，こうして得られた行列により３Ｄベクトルの回転操作をまとめて行い，その後に，投影面上の２本の基本ベクトル（または単位ベクトル）との内積を計算することにより３Ｄ－２Ｄ変換する．それでも，主な計算が行列と３Ｄベクトルの積および３Ｄベクトルの内積になるので，負荷は少し重くなる．

右手座標系か左手座標系のどちらを用いるかの選択は自由であるが，このような工夫をするかしないかにより３Ｄ－２Ｄ変換のための計算量は大幅に異なってくる．




図１７ａ　回転操作により少し傾けた３Ｄ対象物($\alpha$=0, $\beta$=10, $\gamma$=-10,s=0.03)




図１７ｂ　回転操作により少し傾けた３Ｄ対象物($\alpha$=0, $\beta$=10, $\gamma$=-10,s=0.02)

対象物が１個の場合について，平行投影を用いた３Ｄ－２Ｄ変換により，表計算ソフトによる作図機能を用いたコンピュータグラフィックスの例を見てきた．このような作図処理は，今日良く用いられているＣＧの基盤技術を構成している．そして，一般的な表計算ソフトによりこのような背景となる処理を具体例により体験することを通して，３ＤＣＧをより身近に感じることができるものと考えている．

## ４　マクロを用いた作図について

ここで試した計算処理は，見かけ上並列に行われていること，そして回転角度などのデータを入力すると同時にこの値を用いた計算処理が流れるように（データーフロー形式で）行われたことに気付いたであろうか．ＣＧのための多くの処理は並列化が容易になっており，しかも大部分はデータ入力されると即実行できるのが特徴と言える．このデータ入力に関して，各種入力装置やファイルなどを自由に使えるようにすると便利である．

データが揃えば実行されるデータフロー形式による処理の簡単な例として，マウスとキィボードおよび時計を用いるものを紹介しておく．表計算ソフト「ＥＸＣＥＬ」のマクロ機能によりワークシート中に時間を表示し，この表示内容中の秒を用いた計算式により各種変数を１秒毎に計算し，表示されている図を秒単位で変化させるものである．

マクロには，例として以下のものを用いることにする．マクロの名称「DigitalClock」にあるように，ワークシート「３Ｄ螺旋」中のセル「Ｑ１」にデジタル形式「hh:mm:ss」により時間表示をする．変数「TimeValue("00:00:00")」は標準時変更用のオフセットでもあるので省略しても構わない．特定のセル「例としてＳ１」に秒の値（０～５９）だけを取り出す式「=SECOND(Q1)」を書いておくと利用し易くなる．マクロの作成・編集・実行は開発タブからマクロを選んで行う．マクロの動作や停止などの制御をマクロ編集画面からも行えるが，秒の値の表示されているセルを選び，直接に式を書き換え変更したり消去することにより行うのが簡単かもしれない．

```
Sub DigitalClock()
With Sheets("３D 螺旋").Range("Q1")
.Value = Time
.NumberFormatLocal = "hh:mm:ss"
End With
Application.OnTime Now + TimeValue("00:00:00"), "DigitalClock"
End Sub
```

このマクロを実行することによりワークシート「３Ｄ螺旋」中のセル「Ｑ１」にデジタル時計が表示される．他のワークシートからこれを参照し利用するには，式「='３D 螺旋'!Q1　または ='３D 螺旋'!S1」を用いると良い．秒値の表示されているセル「例えばＴ１」を用い，回転角度であれば計算式「例えば=T1*6」などを使うと，データフロー形式の描画システムを操作しているかのようにして，簡単ではあるが動くＣＧを試すことができる．

## ５　複数の線を用いた描画について

散布図を用いて複数の対象物を描くこともできる．簡単な例として，直方体１個を４組の四辺形により描いたものを示す．例では，斜め右上向きが x 軸，斜め左上向きが y 軸，上向きが z 軸になるように表示している．結果（図１８）は，頂点を原点に置いた立方体

を，ｘ軸周りで３０度，ｙ軸周りで１５度回転させたものを平行投影により描いている．



| | A | B | C | D | E | F | G | H | I | J | K | L | M | N |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | π | | | α(z軸回転 | β(y軸回転 | γ(x軸回転) | | | ラジアン=θ×(π/180) | | | |
| 2 | | | 3.14 | | | 0 | 15 | 30 | | | | | | |
| 3 | 頂点の座標 | | | | 2次元平面での座標 | | | | | | | | | 1 |
| 4 | x | y | z | 面 | X | Y1 | Y2 | Y3 | Y4 | | | | | 0 |
| 5 | 0 | 0 | 0 | ① | 0 | 0 | | | | | | | | 0 |
| 6 | 1 | 0 | 0 | ① | 0.965926 | 0.12941 | | | | | | | | |
| 7 | 1 | 0 | 1 | ① | 0.707107 | 0.612372 | | | | | | 0.965926 | | |
| 8 | 0 | 0 | 1 | ① | -0.25882 | 0.482963 | | | | | 横軸: u1 = | 0 | <=Rz(Ry(Rxe1)) | |
| 9 | 0 | 0 | 0 | ① | 0 | 0 | | | | | | -0.25882 | | |
| 10 | 1 | 0 | 0 | ② | 0.965926 | | 0.12941 | | | | | | | |
| 11 | 1 | 1 | 0 | ② | 0.965926 | | 0.995435 | | | | | | | |
| 12 | 1 | 1 | 1 | ② | 0.707107 | | 1.478398 | | | | | 0.12941 | | |
| 13 | 1 | 0 | 1 | ② | 0.707107 | | 0.612372 | | | | 縦軸: u2= | 0.866025 | <=Rz(Ry(Rxe2)) | |
| 14 | 1 | 0 | 0 | ② | 0.965926 | | 0.12941 | | | | | 0.482963 | | |
| 15 | 1 | 1 | 0 | ③ | 0.965926 | | | 0.995435 | | | | | | |
| 16 | 0 | 1 | 0 | ③ | 0 | | | 0.866025 | | | | | | |
| 17 | 0 | 1 | 1 | ③ | -0.25882 | | | 1.348988 | | | | | | |
| 18 | 1 | 1 | 1 | ③ | 0.707107 | | | 1.478398 | | | | | | |
| 19 | 1 | 1 | 0 | ③ | 0.965926 | | | 0.995435 | | | | | | |
| 20 | 0 | 0 | 0 | ④ | 0 | | | | 0 | | | | | |
| 21 | 0 | 0 | 1 | ④ | -0.25882 | | | | 0.482963 | | | | | |
| 22 | 0 | 1 | 1 | ④ | -0.25882 | | | | 1.348988 | | | | | |
| 23 | 0 | 1 | 0 | ④ | 0 | | | | 0.866025 | | | | | |
| 24 | 0 | 0 | 0 | ④ | 0 | | | | 0 | | | | | |



図１８　散布図により直方体を描く

立方体らしいものを構成するため，面①から始めて面④の順に描いているので，最初に描いた辺は後から描く辺により重ね書きされることになる．面④を最後に描くことから，この正方形の全ての辺が同じ色の線で描かれていることを確認することができる．

表計算ソフトによる作図なので，この立方体を構成している４面のデータを表すため，１面毎にセル１列分を用いている．多数の面から成る複雑な対象物を描こうとすると，ワークシートの幅はその数だけ増えて行くことから，修正作業などは煩雑になる．一方、描画用ソフトウェアは，データ表現と管理および加工や再利用などの利便性に配慮している．

なお，CG では表示範囲などを定義するためにビューポート指定をするのが一般的である．しかし，標準設定（軸のオプション＝自動）では全範囲を表示するようになっているため，表示する対象物の向きなどにより軸の比率が変わることになる．これに対し，軸のオプションを固定に変えると，最小値および最大値などを適切に設定する必要はあるものの，ビューポート指定した様な結果が得られることを注記しておく（グラフィクツールー＞レイアウトー＞軸ー＞主横軸などと辿りオプション設定をする）．

## ６　透視投影を用いた簡単な作図について

視点(point of view ;POV)と注目点との２点により注目点を通る投影面を定義する．これより，３Ｄ空間中の視点と対象物上の座標点を通る直線が決まることから，この直線と投

影面との交点すなわち平面上の座標が得られる．この透視変換により，３Ｄ－２Ｄ変換を実行すると，遠近感のある図を作成することができる．以下にその手順の例を示す．

簡単のため，ｘ軸上に視点(d+D,0,0)と注目点(D,0,0)を置くことにする．これは yz 平面に平行な投影面（注目点は x 軸上）を原点から距離 D 離れた位置に設定し，ｘ軸上の視点（観測者）はこの投影面から距離 d だけ離れていることを意味している（図１９）．




図１９　透視投影による座標の求め方

三角形の相似から，３Ｄ対象物上の点(x,y,z)と視点を結ぶ直線と投影面との交点(X,Y,Z)は簡単な計算により求めることができ，これらからｙｚ平面上の２Ｄ座標(Y,Z)を得る．

$$X=D,\quad Y=y*d/(d+D-x),\quad Z=z*d/(d+D-x),$$

透視投影を用いたこの３Ｄ－２Ｄ変換では，先ず視点(d+D,0,0)と注目点(D,0,0)とによりｙｚ平面に平行な投影面を定義している．次に，３Ｄ回転行列の積 RzRyRx を計算し，この行列を用いて回転操作をした対象物の３Ｄ座標点を求める．最後に，こうして得られた３Ｄ座標点のｙｚ座標成分から２Ｄ座標(Y,Z)を求める．以下にその手順を示す．

座標軸周りの回転を表す行列の積を順次求める．行列(Ry)(Rx)の各成分は次のようにな る．



S8　=$S3*W$3+$T3*W$4+$U3*W$5

| | N | O | P | Q | R | S | T | U | V | W | X | Y |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | | | | | | | | | | |
| 2 | | | Rz | | | | Ry | | | | Rx | |
| 3 | | 0.866 | -0.5 | 0 | | 1 | 0 | 0 | | 1 | 0 | 0 |
| 4 | | 0.5 | 0.866 | 0 | | 0 | 1 | 0 | | 0 | 1 | 0 |
| 5 | | 0 | 0 | 1 | | 0 | 0 | 1 | | 0 | 0 | 1 |
| 6 | | | | | | | | | | | | |
| 7 | | | Rz(RyRx) | | | | RyRx | | | | | |
| 8 | | 0.866 | -0.5 | 0 | | 1 | 0 | 0 | | | | |
| 9 | | 0.5 | 0.866 | 0 | | 0 | 1 | 0 | | | | |
| 10 | | 0 | 0 | 1 | | 0 | 0 | 1 | | | | |

図２０　回転操作を表す行列(Ry)(Rx)を計算

式「=$S3*W$3+$T3*W$4+$U3*W$5　，・・・，=$S5*W$3+$T5*W$4+$S*W$5」により行列の１行１列から３行１列までの成分を計算する．同様にして，１行２列から３行３列までを式

「=$S3*X$3+$T3*X$4+$U3*X$5　，・・・，=$S5*Y$3+$T5*Y$4+$U5*Y$5」により得る．

O8　=$O3*S$8+$P3*S$9+$Q3*S$10

| | N | O | P | Q | R | S | T | U | V | W | X | Y |
|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | | | | | | | | | | | | |
| 2 | | | Rz | | | | Ry | | | | Rx | |
| 3 | | 0.866 | −0.5 | 0 | | 1 | 0 | 0 | | 1 | 0 | 0 |
| 4 | | 0.5 | 0.866 | 0 | | 0 | 1 | 0 | | 0 | 1 | 0 |
| 5 | | 0 | 0 | 1 | | 0 | 0 | 1 | | 0 | 0 | 1 |
| 6 | | | | | | | | | | | | |
| 7 | | | Rz(RyRx) | | | | RyRx | | | | | |
| 8 | | 0.866 | −0.5 | 0 | | 1 | 0 | 0 | | | | |
| 9 | | 0.5 | 0.866 | 0 | | 0 | 1 | 0 | | | | |
| 10 | | 0 | 0 | 1 | | 0 | 0 | 1 | | | | |

図２１　回転操作を表す行列(Rz)(Ry)(Rx)を計算

次に，式「=$O3*S$8+$P3*S$9+$Q3*S$10　，・・・，=$O5*U$8+$P5*U$9+$Q5*U$10」により，回転操作に用いる行列(Rz)(Ry)(Rx)の１行１列から３行３列までの成分を得る（図１９）．

こうして得られた行列(Rz)(Ry)(Rx)を用い，対象物の３Ｄ座標(xp, yp, zp)から回転操作後の３Ｄ座標(x, y, z)を計算する．ｘ座標「Ｅ５」を式「=$O$8*$A5+$P$8*$B5+$Q$8*$C5+$J$2」により計算する．同様に，ｙ座標とｚ座標の値を式「=$O$9*$A5+$P$9*$B5+$Q$9*$C5+$K$2　と　=$O$10*$A5+$P$10*$B5+$Q$10*$C5+$L$2」により計算する（図２２・・・X, Y1, Y2, Y3, Y4）．

ただし式の中にある最後の項目，セル「$J$2」はｘ軸方向，セル「$K$2」はｙ軸方向，セル「$L$2」はｚ軸方向に関する平行移動を表している．

E5　=$O$8*$A5+$P$8*$B5+$Q$8*$C5+$J$2

| | A | B | C | D | E | F | G | H | I | J | K | L | M | N | O | P | Q |
|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|---|
| 1 | R | | π | | α(z軸周) | β(y軸周) | γ(x軸周) | D | d | a(x軸) | b(y軸) | c(z軸) | | | | | |
| 2 | 1 | | 3.14 | | 30 | 0 | 0 | 2 | 3 | 0 | 0 | 0 | | | | Rz | |
| 3 | 頂点の座標 | | | | | | | | | | | | | | 0.866 | −0.5 | 0 |
| 4 | xp | yp | zp | 面 | x | y | z | X | Y1 | Y2 | Y3 | Y4 | Y5 | | 0.5 | 0.866 | 0 |
| 5 | −1 | −1 | −1 | ① | −0.366 | −1.366 | −1 | −0.764 | −0.559 | | | | | | 0 | 0 | 1 |
| 6 | 1 | −1 | −1 | ① | 1.36603 | −0.366 | −1 | −0.302 | −0.826 | | | | | | | | |
| 7 | 1 | −1 | 1 | ① | 1.36603 | −0.366 | 1 | −0.302 | 0.8255 | | | | | | | Rz(RyRx) | |
| 8 | −1 | −1 | 1 | ① | −0.366 | −1.366 | 1 | −0.764 | 0.5591 | | | | | | 0.866 | −0.5 | 0 |
| 9 | −1 | −1 | −1 | ① | −0.366 | −1.366 | −1 | −0.764 | −0.559 | | | | | | 0.5 | 0.866 | 0 |
| 10 | 1 | −1 | −1 | ② | 1.36603 | −0.366 | −1 | −0.302 | | −0.826 | | | | | 0 | 0 | 1 |
| 11 | 1 | 1 | −1 | ② | 0.36603 | 1.36603 | −1 | 0.8844 | | −0.647 | | | | | | | |
| 12 | 1 | 1 | 1 | ② | 0.36603 | 1.36603 | 1 | 0.8844 | | 0.6474 | | | | | | | |
| 13 | 1 | −1 | 1 | ② | 1.36603 | −0.366 | 1 | −0.302 | | 0.8255 | | | | | | | |
| 14 | 1 | −1 | −1 | ② | 1.36603 | −0.366 | −1 | −0.302 | | −0.826 | | | | | | | |
| 15 | 1 | 1 | −1 | ③ | 0.36603 | 1.36603 | −1 | 0.8844 | | | −0.647 | | | | | | |
| 16 | −1 | 1 | −1 | ③ | −1.366 | 0.36603 | −1 | 0.1725 | | | −0.471 | | | | | | |
| 17 | −1 | 1 | 1 | ③ | −1.366 | 0.36603 | 1 | 0.1725 | | | 0.4713 | | | | | | |
| 18 | 1 | 1 | 1 | ③ | 0.36603 | 1.36603 | 1 | 0.8844 | | | 0.6474 | | | | | | |
| 19 | 1 | 1 | −1 | ③ | 0.36603 | 1.36603 | −1 | 0.8844 | | | −0.647 | | | | | | |
| 20 | −1 | −1 | −1 | ④ | −0.366 | −1.366 | −1 | −0.764 | | | | −0.559 | | | | | |
| 21 | −1 | −1 | 1 | ④ | −0.366 | −1.366 | 1 | −0.764 | | | | 0.5591 | | | | | |
| 22 | −1 | 1 | 1 | ④ | −1.366 | 0.36603 | 1 | 0.1725 | | | | 0.4713 | | | | | |
| 23 | −1 | 1 | −1 | ④ | −1.366 | 0.36603 | −1 | 0.1725 | | | | −0.471 | | | | | |
| 24 | −1 | −1 | −1 | ④ | −0.366 | −1.366 | −1 | −0.764 | | | | −0.559 | | | | | |
| 25 | 1 | 0 | 1 | ⑤ | 0.86603 | 0.5 | 1 | 0.3628 | | | | | 0.7257 | | | | |

図２２　回転操作をした３Ｄ座標の計算

先に触れたように，行列の各成分の計算については，(1,1)成分の計算式をコピーすることで残り８成分が求まるよう，絶対参照と相対参照表記とを混在させている．相対参照のみによると，同じ結果は得られるのものの，かなり非効率になることを注意しておく．

このような３Ｄ－２Ｄ変換処理により，作図することができる．３Ｄ対象物の例として辺の長さが２の立方体を選び，中心を原点に，各辺を座標軸と平行になるように置き，注目点(D=2,0,0)と注視点(d=3,0,0)を決め，透視投影をする．回転操作の例として，ｘ軸周りで３０度回転させたものを示す．なお，１２角形を追加し，軸の書式設定メニュー「軸のオプション，線の色」などから座標軸や目盛を描かないよう設定している（図２３）．




図２３　立方体を平行移動または回転操作して透視投影

## 7　レポート課題

指示された課題に従った内容のレポートを作成してください．

（１）座標が簡単な式により与えられる２Ｄ図形を描き解説せよ．

（２）座標が簡単な式により与えられる３Ｄ図形を描き解説せよ．



図２４　各種図形

メモ：この資料は商業目的外なら，出典を明記するだけで自由に利用することができる．